

Examining BP and WP Effects of Negative Mood Predicting Next-Morning Glucose COMPLETED VERSION

These data were simulated loosely based on real data reported in the citation below. The daily diary study followed persons with Type II diabetes for 21 consecutive days to examine within-person relationships between mood, stress, and morning glucose (an index of how well-controlled the diabetes is). Here we will examine between-person and within-person relationships between daily negative mood and glucose the next morning (which was log-transformed given skewness) and how these relationships are moderated by sex.

Skaff, M., Mullan J., Fisher, L., Almeida, D., **Hoffman, L.**, Masharani, U., & Mohr, D. (2009). [Effects of mood on daily fasting glucose in Type 2 Diabetes](#). *Health Psychology*, 28(3), 265-272.

SAS Data Setup:

```
* SAS code to read data into work library and center predictors;
DATA work.example6; SET filepath.example6;
* Level-2 effect of Negative Mood (mean=0, SD=1);
  PMnm0 = PMnegmood - 0; LABEL PMnm0= "PMnm0: Person Mean Negative Mood (0=0)";
* Level-1 effect to use with PERSON-MEAN-CENTERING;
  WPnm = negmood - PMnegmood; LABEL wpnm= "WPnm: Within-Person Negative Mood (0=PM)";
* Level-1 effect to use with GRAND-MEAN-CENTERING;
  TVnm0 = negmood - 0; LABEL TVnm0= "TVnm0: Time-Varying Negative Mood (0=0)";
RUN;
```

SPSS Data Setup:

```
* SPSS code to import data and center predictors.
GET FILE = "example/Example6.sav".
DATASET NAME example6 WINDOW=FRONT.
COMPUTE PMnm0 = PMnegmood - 0.
COMPUTE WPnm = negmood - PMnegmood.
COMPUTE TVnm0 = negmood - 0.
VARIABLE LABELS
  PMnm0 "PMnm0: Person Mean Negative Mood (0=0)"
  WPnm "WPnm: Within-Person Negative Mood (0=PM)"
  TVnm0 "TVnm0: Time-Varying Negative Mood (0=0)".
EXECUTE.
```

STATA Data Setup:

```
* STATA code to center predictors
* level-2 effect of negative mood
gen PMnm0 = PMnegmood - 0
label variable PMnm0 "PMnm0: Person Mean Negative Mood (0=0)"
* level-1 effect to use with PERSON-MEAN-CENTERING
gen WPnm = negmood - PMnegmood
label variable WPnm "WPnm: Within-Person Negative Mood (0=PM)"
* level-1 effect to use with GRAND-MEAN-CENTERING
gen TVnm0 = negmood - 0
label variable TVnm0 "TVnm0: Time-Varying Negative Mood (0=0)"
* interaction terms needed for lincom statements
gen WPnmsexmf = WPnm * sexmf
gen PMnm0sexmf = PMnm0 * sexmf
gen TVnm0sexmf = TVnm0 * sexmf
```

Model 1a. Empty Model for LN Morning Glucose (Daily Outcome)

```

TITLE "SAS Model 1a: Empty Model for Daily Glucose Outcome";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
  CLASS ID day;
  MODEL lglucAM = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / VCORR SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC; RUN;

TITLE "SPSS Model 1a: Empty Model for Daily Glucose Outcome".
MIXED lglucAM BY ID day
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV
  /FIXED =
  /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = day | SUBJECT(ID) COVTYPE(ID).

* STATA Model 1a: Empty Model for Daily Glucose Outcome
xtmixed lglucAM , || id: , variance ml covariance(un) residuals(independent,t(day))

```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	0.06654	0.006690	9.95	<.0001
day	ID	0.03029	0.000683	44.35	<.0001

ICC for Glucose:

$$.06654 / (.06654 + .03029) = .6872$$

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1941.5	3	-1935.5	-1935.5	-1931.4	-1925.5	-1922.5

Model 1b. Empty Model for Negative Mood (Daily Predictor)

```

TITLE "SAS Model 1b: Empty Model for Daily Negative Mood Predictor";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
  CLASS ID day;
  MODEL negmood = / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / VCORR SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC; RUN;

TITLE "SPSS Model 1b: Empty Model for Daily Negative Mood Predictor".
MIXED negmood BY ID day
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV
  /FIXED =
  /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = day | SUBJECT(ID) COVTYPE(ID).

* STATA Model 1b: Empty Model for Daily Negative Mood Predictors
xtmixed negmood , || id: , ///
  variance ml covariance(un) residuals(independent,t(day))

```

Estimated V Correlation Matrix for ID 1										
Row	Col1	Col2	Col3	Col4	Col5	Col6	Col7	Col8	Col9	Col10
1	1.0000	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895
2	0.3895	1.0000	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895
3	0.3895	0.3895	1.0000	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895	0.3895

.....

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	0.3355	0.03557	9.43	<.0001
day	ID	0.5258	0.01186	44.35	<.0001

ICC for Negative Mood:

$$.3355 / (.3353 + .5258) = .3895$$

Model 2a. Fixed Effects of Negative Mood using Person-Mean-Centering (PMC)

```
TITLE "SAS Model 2a: Fixed Effects of Negative Mood using PMC";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
  CLASS ID day;
  MODEL lglucAM = WPnm PMnm0 / SOLUTION DDFM=Satterthwaite OUTPM=MoodPred;
  RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC;
  ESTIMATE "Within-Person Mood Effect"      WPnm 1;
  ESTIMATE "Between-Person Mood Effect"     PMnm0 1;
  ESTIMATE "Contextual Mood Effect"         PMnm0 1 WPnm -1;
RUN; PROC CORR NOSIMPLE DATA=MoodPred; VAR lglucAM pred; RUN;
```

```
TITLE "SPSS Model 2a: Fixed Effects of Negative Mood using PMC".
MIXED lglucAM BY ID day WITH WPnm PMnm0
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV
  /FIXED = WPnm PMnm0
  /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = day | SUBJECT(ID) COVTYPE(ID)
  /SAVE = FIXPRED (predmood)
  /TEST = "Within-Person Mood Effect"      WPnm 1
  /TEST = "Between-Person Mood Effect"     PMnm0 1
  /TEST = "Contextual Mood Effect"         PMnm0 1 WPnm -1.
CORRELATIONS lglucAM predmood.
```

```
* STATA Model 2a: Fixed Effects of Negative Mood using PMC
xtmixed lglucAM c.WPnm c.PMnm0, || id: , ///
  variance ml covariance(un) residuals(independent,t(day)),
  estat ic, n(207),
  predict predmood,
  estimates store FixWP,
  lincom 1*WPnm          // within-person mood effect
  lincom 1*PMnm0         // between-person mood effect
  lincom 1*PMnm0 - 1*WPnm // contextual mood effect
corr lglucAM predmood
```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	0.06435	0.006474	9.94	<.0001 → random intercept variance reduced 3.29%
day	ID	0.03022	0.000682	44.35	<.0001 → residual variance reduced 0.23%

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1956.5	5	-1946.5	-1946.5	-1939.8	-1929.9	-1924.9

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	4.9302	0.01845	207	267.20	<.0001
WPnm	0.01097	0.003821	3941	2.87	0.0041
PMnm0	0.08040	0.03046	207	2.64	0.0089

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
Within-Person Mood Effect	0.01097	0.003821	3941	2.87	0.0041
Between-Person Mood Effect	0.08040	0.03046	207	2.64	0.0089
Contextual Mood Effect	0.06942	0.03070	213	2.26	0.0247

Pearson Correlation Coefficients, N = 4140

	lglucAM	Pred
lglucAM	1.00000	0.15269
lglucAM: Log Morning Glucose		<.0001

Total R² from mood = .023

Is this a better model than the empty model (1a)—is the mood total R^2 significant?

Yes, $ML -2\Delta LL(2) = 11.0, p = .004$

What does the level-1 effect (WPnm) represent in this model?

The level-1 effect is the within-person effect of negative mood. For every unit relative increase in your own negative mood that day, that next day's glucose goes up by .01097 (WP relation among daily levels).

What does the level-2 effect (PMnm0) represent in this model?

The level-2 effect is the between-person effect of negative mood. For every unit higher person mean negative mood, mean glucose is higher by .01097 (BP relation among mean levels).

What does the “contextual mood effect” represent?

It is the difference in the between-person and within-person effects: the between-person mood effect is significantly greater than the within-person mood effect by .069 (so convergence was not obtained). In other words, after controlling for current negative mood, there is an incremental effect of .069 per unit higher person mean negative mood.

Model 2b. Random Effect of WP Negative Mood under PMC

```
TITLE "SAS Model 2b: Random Effect of WP Negative Mood using PMC";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
  CLASS ID day;
  MODEL lglucAM = WPnm PMnm0 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT WPnm / SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC;
  ESTIMATE "Within-Person Mood Effect"      WPnm 1;
  ESTIMATE "Between-Person Mood Effect"     PMnm0 1;
  ESTIMATE "Contextual Mood Effect"         PMnm0 1 WPnm -1; RUN;
```

```
TITLE "SPSS Model 2b: Random Effect of WP Negative Mood using PMC".
```

```
MIXED lglucAM BY ID day WITH WPnm PMnm0
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV
  /FIXED = WPnm PMnm0
  /RANDOM = INTERCEPT WPnm | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = day | SUBJECT(ID) COVTYPE(ID)
  /TEST = "Within-Person Mood Effect"      WPnm 1
  /TEST = "Between-Person Mood Effect"     PMnm0 1
  /TEST = "Contextual Mood Effect"         PMnm0 1 WPnm -1.
```

```
* STATA Model 2b: Random Effect of WP Negative Mood using PMC
```

```
xtmixed lglucAM c.WPnm c.PMnm0, || id: WPnm, ///
  variance ml covariance(un) residuals(independent,t(day)),
  estat ic, n(207),
  estimates store RandWP,
  lrtest RandWP FixWP,
  lincom 1*WPnm           // within-person mood effect
  lincom 1*PMnm0          // between-person mood effect
  lincom 1*PMnm0 - 1*WPnm // contextual mood effect
```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	0.06440	0.006479	9.94	<.0001 random intercept variance
UN(2,1)	ID	-0.00020	0.001067	-0.19	0.8478 intercept-WP slope covariance
UN(2,2)	ID	0.000505	0.000335	1.51	0.0656 random WP mood slope variance
day	ID	0.02995	0.000692	43.28	<.0001 residual variance

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1959.4	7	-1945.4	-1945.4	-1936.0	-1922.1	-1915.1

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	4.9302	0.01846	207	267.10	<.0001
WPnm	0.01104	0.004137	202	2.67	0.0083
PMnm0	0.08022	0.03047	207	2.63	0.0091

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
Within-Person Mood Effect	0.01104	0.004137	202	2.67	0.0083
Between-Person Mood Effect	0.08022	0.03047	207	2.63	0.0091
Contextual Mood Effect	0.06918	0.03075	215	2.25	0.0255

Is this a better model than the fixed effects model (2a)? How do we know? What does this result mean?

No, $ML - 2\Delta LL(2) = 2.91, p = .235$

It means that so far, each person does not need his or her own effect of worse negative mood than usual.

Model 2c. Adding Moderation Effects by Sex (0=M, 1=F) for Each Mood Effect under PMC

```

TITLE "SAS Model 2c: Fixed Effects of Sex (0=M, 1=F) by PMC Negative Mood";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
  CLASS ID day;
  MODEL lglucAM = WPnm PMnm0 sexmf WPnm*sexmf PMnm0*sexmf
    / SOLUTION DDFM=Satterthwaite OUTPM=SexPred;
  RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC;
ESTIMATE "Intercept: Men (Mood=0)"          intercept 1 sexmf 0;
ESTIMATE "Intercept: Women (Mood=0)"         intercept 1 sexmf 1;
ESTIMATE "Intercept: Women Diff (Mood=0)"    sexmf 1;
ESTIMATE "Within-Person Mood Effect: Men"     WPnm 1 WPnm*sexmf 0;
ESTIMATE "Within-Person Mood Effect: Women"   WPnm 1 WPnm*sexmf 1;
ESTIMATE "Within-Person Mood Effect: Women Diff" WPnm*sexmf 1;
ESTIMATE "Between-Person Mood Effect: Men"     PMnm0 1 PMnm0*sexmf 0;
ESTIMATE "Between-Person Mood Effect: Women"   PMnm0 1 PMnm0*sexmf 1;
ESTIMATE "Between-Person Mood Effect: Women Diff" PMnm0*sexmf 1;
ESTIMATE "Contextual Mood Effect: Men"         PMnm0 1 PMnm0*sexMF 0 WPnm -1 WPnm*sexMF 0;
ESTIMATE "Contextual Mood Effect: Women"       PMnm0 1 PMnm0*sexMF 1 WPnm -1 WPnm*sexMF -1;
ESTIMATE "Contextual Mood Effect: Women Diff" PMnm0*sexMF 1 WPnm*sexMF -1;
RUN; PROC CORR NOSIMPLE DATA=SexPred; VAR lglucAM pred; RUN;

TITLE "SPSS Model 2c: Fixed Effects of Sex (0=M, 1=F) by PMC Negative Mood".
MIXED lglucAM BY ID day WITH WPnm PMnm0 sexmf
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV
  /FIXED = WPnm PMnm0 sexmf WPnm*sexmf PMnm0*sexmf
  /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = day | SUBJECT(ID) COVTYPE(ID)
  /SAVE = FIXPRED (predsex)
/TEST = "Intercept: Men (Mood=0)"          intercept 1 sexmf 0
/TEST = "Intercept: Women (Mood=0)"         intercept 1 sexmf 1
/TEST = "Intercept: Women Diff (Mood=0)"    sexmf 1
/TEST = "Within-Person Mood Effect: Men"     WPnm 1 WPnm*sexmf 0
/TEST = "Within-Person Mood Effect: Women"   WPnm 1 WPnm*sexmf 1
/TEST = "Within-Person Mood Effect: Women Diff" WPnm*sexmf 1
/TEST = "Between-Person Mood Effect: Men"     PMnm0 1 PMnm0*sexmf 0
/TEST = "Between-Person Mood Effect: Women"   PMnm0 1 PMnm0*sexmf 1
/TEST = "Between-Person Mood Effect: Women Diff" PMnm0*sexmf 1
/TEST = "Contextual Mood Effect: Men"         PMnm0 1 PMnm0*sexMF 0 WPnm -1 WPnm*sexMF 0
/TEST = "Contextual Mood Effect: Women"       PMnm0 1 PMnm0*sexMF 1 WPnm -1 WPnm*sexMF -1
/TEST = "Contextual Mood Effect: Women Diff" PMnm0*sexMF 1 WPnm*sexMF -1.
CORRELATIONS lglucAM predsex.

```

```

* STATA Model 2c: SPSS Model 2c: Fixed Effects of Sex (0=M, 1=F) by PMC Negative Mood
xtmixed lglucAM c.WPnm c.PMnm0 c.sexmf c.WPnmsexmf c.PMnm0sexmf, ///
|| id: , variance ml covariance(un) residuals(independent,t(day)),
estat ic, n(207),
predict predsex,
lincom 1*_cons + 0*sexmf // intercept: men (mood=0)
lincom 1*_cons + 1*sexmf // intercept: women (mood=0)
lincom 1*sexmf // intercept: women diff (mood=0)
lincom 1*WPnm + 0*WPnmsexmf // within-person mood effect: men
lincom 1*WPnm + 1*WPnmsexmf // within-person mood effect: women
lincom 1*WPnmsexmf // within-person mood effect: women diff
lincom 1*PMnm0 + 0*PMnm0sexmf // between-person mood effect: men
lincom 1*PMnm0 + 1*PMnm0sexmf // between-person mood effect: women
lincom 1*PMnm0sexmf // between-person mood effect: women diff
lincom 1*PMnm0 + 0*PMnm0sexmf - 1*WPnm + 0*WPnmsexmf // contextual mood effect: men
lincom 1*PMnm0 + 1*PMnm0sexmf - 1*WPnm - 1*WPnmsexmf // contextual mood effect: women
lincom 1*PMnm0sexmf - 1*WPnm sexmf // contextual mood effect: women diff
corr lglucAM predsex

```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	0.06074	0.006118	9.93	<.0001 → random intercept variance reduced 5.61%
day	ID	0.03007	0.000678	44.35	<.0001 → residual variance reduced 0.50%

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1988.1	8	-1972.1	-1972.0	-1961.3	-1945.4	-1937.4

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	4.9539	0.02734	207	181.21	<.0001
WPnm	0.03119	0.005937	3942	5.25	<.0001
PMnm0	0.1996	0.04849	207	4.12	<.0001
sexmf	-0.03619	0.03626	207	-1.00	0.3194
WPnm*sexmf	-0.03443	0.007743	3942	-4.45	<.0001
PMnm0*sexmf	-0.1849	0.06135	207	-3.01	0.0029

Is this a better model than the PMC mood-only model 2a?

Yes, $ML -2\Delta LL(3) = 31.6, p < .001$

What does the intercept now represent in this model?

The intercept of 4.9539 is the expected glucose for a man with a $PMnm = 0$ and $WPnm = 0$.

What does the level-1 effect (WPnm) represent in this model?

The level-1 effect is the simple within-person effect of negative mood specifically for a man. For every unit relative increase in your own negative mood that day, that next day's glucose goes up by 0.03119 (significant).

What does the level-2 effect (PMnm0) represent in this model?

The level-2 effect is the simple between-person effect of negative mood specifically for a man. For every unit increase in your person mean negative mood, mean glucose is higher by 0.1996 (significant).

What does the main effect of sex represent in this model?

The simple effect of sex is the difference between men and women for someone with a person mean negative mood of 0 on day when they are at their mean. In those persons, women are -0.03619 lower in mean glucose (n.s.).

What does the WPnm*Sex interaction represent in this model?

The $WP*Sex$ interaction tells us that the WP mood effect is 0.03443 smaller in women (significant interaction).

What does the PMnm0*Sex interaction represent in this model?

The $BP*Sex$ interaction tells us BP mood effect is 0.1849 smaller in women (significant interaction).

Label	Estimates		DF	t Value	Pr > t
	Estimate	Standard Error			
Intercept: Men (Mood=0)	4.9539	0.02734	207	181.21	<.0001
Intercept: Women (Mood=0)	4.9177	0.02382	207	206.42	<.0001
Intercept: Women Diff (Mood=0)	-0.03619	0.03626	207	-1.00	0.3194
Within-Person Mood Effect: Men	0.03119	0.005937	3942	5.25	<.0001
Within-Person Mood Effect: Women	-0.00325	0.004970	3942	-0.65	0.5138
Within-Person Mood Effect: Women Diff	-0.03443	0.007743	3942	-4.45	<.0001
Between-Person Mood Effect: Men	0.1996	0.04849	207	4.12	<.0001
Between-Person Mood Effect: Women	0.01469	0.03759	207	0.39	0.6962
Between-Person Mood Effect: Women Diff	-0.1849	0.06135	207	-3.01	0.0029
Contextual Mood Effect: Men	0.1684	0.04886	214	3.45	0.0007
Contextual Mood Effect: Women	0.01794	0.03790	214	0.47	0.6364
Contextual Mood Effect: Women Diff	-0.1505	0.06184	214	-2.43	0.0158

Which of these effects are not directly given by the model?

The effects for women and all of the contextual effects


Pearson Correlation Coefficients, N = 4140

Prob > |r| under H0: Rho=0

	lgLucAM	Pred
lgLucAM	1.00000	0.24931
lgLucAM: Log Morning Glucose		<.0001

Total R^2 from mood+sex = .062, for a net increase of .039 from sex effects

Figure 1:



Model 3. Predicting Glucose from Time-Varying Negative Mood only (GMC):

```

TITLE "SAS Model 3: Fixed Effect of TV Negative Mood only using GMC";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
  CLASS ID day;
  MODEL lglucAM = TVnm0 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC; RUN;

TITLE "SPSS Model 3: Fixed Effect of TV Negative Mood only using GMC".
MIXED lglucAM BY ID day WITH TVnm0
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV
  /FIXED = TVnm0
  /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = day | SUBJECT(ID) COVTYPE(ID).

* STATA Model 3: Fixed Effect of TV Negative Mood only using GMC
xtmixed lglucAM c.TVnm0, || id: , variance ml covariance(un) residuals(independent,t(day)),
  estat ic, n(207)

```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	0.06595	0.006634	9.94	<.0001 → random intercept variance reduced 0.89%
day	ID	0.03022	0.000682	44.34	<.0001 → residual variance reduced 0.23%

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1951.5	4	-1943.5	-1943.5	-1938.1	-1930.2	-1926.2

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	4.9408	0.01806	207	273.52	<.0001
TVnm0	0.01202	0.003792	4041	3.17	0.0015

What does the effect of TVnm0 represent in this model?

It is the smushed (conflated, convergence) effect of mood.

Model 3a. Fixed Effects of Negative Mood using Grand-Mean-Centering (GMC)

```

TITLE "SAS Model 3a: Fixed Effects of Negative Mood using GMC";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
  CLASS ID day;
  MODEL lglucAM = TVnm0 PMnm0 / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC;
  ESTIMATE "Within-Person Mood Effect" TVnm0 1;
  ESTIMATE "Between-Person Mood Effect" TVnm0 1 PMnm0 1;
  ESTIMATE "Contextual Mood Effect" PMnm0 1; RUN;

TITLE "SPSS Model 3a: Fixed Effects of Negative Mood using GMC".
MIXED lglucAM BY ID day WITH TVnm0 PMnm0
  /METHOD = ML
  /PRINT = SOLUTION TESTCOV
  /FIXED = TVnm0 PMnm0
  /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
  /REPEATED = day | SUBJECT(ID) COVTYPE(ID)
  /TEST = "Within-Person Mood Effect" TVnm0 1
  /TEST = "Between-Person Mood Effect" TVnm0 1 PMnm0 1
  /TEST = "Contextual Mood Effect" PMnm0 1.

```



```
* STATA Model 3a: Fixed Effects of Negative Mood using GMC
xtmixed lglucAM c.TVnm0 c.PMnm0, || id: , ///
variance ml covariance(un) residuals(independent,t(day)),
estat ic, n(207),
estimates store FixTV,
lincom 1*TVnm0 // within-person mood effect
lincom 1*TVnm0 + 1*PMnm0 // between-person mood effect
lincom 1*PMnm0 // contextual mood effect
```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z
UN(1,1)	ID	0.06435	0.006474	9.94	<.0001 → random intercept further reduced 2.43%
day	ID	0.03022	0.000682	44.35	<.0001 → residual variance not reduced

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1956.5	5	-1946.5	-1946.5	-1939.8	-1929.9	-1924.9

Note that the fit is the same as model 2a (and thus the R² values are, too)

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	4.9302	0.01845	207	267.20	<.0001
TVnm0	0.01097	0.003821	3941	2.87	0.0041
PMnm0	0.06942	0.03070	213	2.26	0.0247

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
Within-Person Mood Effect	0.01097	0.003821	3941	2.87	0.0041
Between-Person Mood Effect	0.08040	0.03046	207	2.64	0.0089
Contextual Mood Effect	0.06942	0.03070	213	2.26	0.0247

What does the level-2 effect (PMnm0) represent in this model?

It is the difference in the between-person and within-person effects (the contextual effect): the between-person mood effect is significantly greater than the within-person mood effect by .069 (so convergence was not obtained).

In other words, after controlling for current negative mood, there is an incremental effect of .069 per unit higher person mean negative mood.

Model 3b. Random Effect of TV Negative Mood under GMC

```
TITLE "SAS Model 3b: Random Effect of TV Negative Mood using GMC";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
CLASS ID day;
MODEL lglucAM = TVnm0 PMnm0 / SOLUTION DDFM=Satterthwaite;
RANDOM INTERCEPT TVnm0 / SUBJECT=ID TYPE=UN;
REPEATED day / SUBJECT=ID TYPE=VC;
ESTIMATE "Within-Person Mood Effect" TVnm0 1;
ESTIMATE "Between-Person Mood Effect" TVnm0 1 PMnm0 1;
ESTIMATE "Contextual Mood Effect" PMnm0 1; RUN;
```

```
TITLE "SPSS Model 3b: Random Effect of TV Negative Mood using GMC".
MIXED lglucAM BY ID day WITH TVnm0 PMnm0
/METHOD = ML
/PRINT = SOLUTION TESTCOV
/FIXED = TVnm0 PMnm0
/RANDOM = INTERCEPT TVnm0 | SUBJECT(ID) COVTYPE(UN)
/REPEATED = day | SUBJECT(ID) COVTYPE(ID)
/TEST = "Within-Person Mood Effect" TVnm0 1
/TEST = "Between-Person Mood Effect" TVnm0 1 PMnm0 1
/TEST = "Contextual Mood Effect" PMnm0 1.
```

```

* STATA Model 3b: Random Effect of WP Negative Mood using GMC
xtmixed lglucAM c.TVnm0 c.PMnm0, || id: TVnm0, ///
variance ml covariance(un) residuals(independent,t(day)),
estat ic, n(207),
estimates store RandTV,
lrtest RandTV FixTV,
lincom 1*TVnm0 // within-person mood effect
lincom 1*TVnm0 + 1*PMnm0 // between-person mood effect
lincom 1*PMnm0 // contextual mood effect

```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	0.06400	0.006464	9.90	<.0001 random intercept variance
UN(2,1)	ID	-0.00033	0.001050	-0.31	0.7549 int-TV mood slope covariance
UN(2,2)	ID	0.000579	0.000339	1.71	0.0441 TV mood slope variance
day	ID	0.02992	0.000690	43.34	<.0001 residual variance

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1960.4	7	-1946.4	-1946.4	-1937.0	-1923.1	-1916.1

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	4.9302	0.01843	206	267.45	<.0001
TVnm0	0.01102	0.004181	205	2.64	0.0090
PMnm0	0.07015	0.03066	214	2.29	0.0231

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
Within-Person Mood Effect	0.01102	0.004181	205	2.64	0.0090
Between-Person Mood Effect	0.08117	0.03047	209	2.66	0.0083
Contextual Mood Effect	0.07015	0.03066	214	2.29	0.0231

Is this a better model than the fixed effects model (3a)? How do we know? What does this result mean?

No, $ML - 2\Delta LL(2) = 3.9, p = .142$

It means that so far, each person does not need his or her own effect of worse negative mood than usual.

Model 3c. Adding Moderation Effects by Sex (0=M, 1=F) for Each Mood Effect under GMC

```

TITLE "SAS Model 3c: Fixed Effects of Sex (0=M, 1=F) by GMC Negative Mood";
PROC MIXED DATA=work.example6 COVTEST NOCLPRINT NOITPRINT NAMELEN=100 IC METHOD=ML;
  CLASS ID day;
  MODEL lglucAM = TVnm0 PMnm0 sexmf TVnm0*sexmf PMnm0*sexmf
    / SOLUTION DDFM=Satterthwaite;
  RANDOM INTERCEPT / SUBJECT=ID TYPE=UN;
  REPEATED day / SUBJECT=ID TYPE=VC;
ESTIMATE "Intercept: Men (Mood=0)" intercept 1 sexmf 0;
ESTIMATE "Intercept: Women (Mood=0)" intercept 1 sexmf 1;
ESTIMATE "Intercept: Women Diff (Mood=0)" sexmf 1;
ESTIMATE "Within-Person Effect: Men" TVnm0 1 TVnm0*sexmf 0;
ESTIMATE "Within-Person Effect: Women" TVnm0 1 TVnm0*sexmf 1;
ESTIMATE "Within-Person Effect: Women Diff" TVnm0*sexmf 1;
ESTIMATE "Between-Person Effect: Men" TVnm0 1 TVnm0*sexmf 0 PMnm0 1 PMnm0*sexmf 0;
ESTIMATE "Between-Person Effect: Women" TVnm0 1 TVnm0*sexmf 1 PMnm0 1 PMnm0*sexmf 1;
ESTIMATE "Between-Person Effect: Women Diff" TVnm0*sexmf 1 PMnm0*sexmf 1;
ESTIMATE "Contextual Effect: Men" PMnm0 1 PMnm0*sexmf 0;
ESTIMATE "Contextual Effect: Women" PMnm0 1 PMnm0*sexmf 1;
ESTIMATE "Contextual Effect: Women Diff" PMnm0*sexmf 1; RUN;

```

```

TITLE "SPSS Model 3c: Fixed Effects of Sex (0=M, 1=F) by GMC Negative Mood".
MIXED lglucAM BY ID day WITH TVnm0 PMnm0 sexmf
    /METHOD = ML
    /PRINT = SOLUTION TESTCOV
    /FIXED = TVnm0 PMnm0 sexmf TVnm0*sexmf PMnm0*sexmf
    /RANDOM = INTERCEPT | SUBJECT(ID) COVTYPE(UN)
    /REPEATED = day | SUBJECT(ID) COVTYPE(ID)
/TEST = "Intercept: Men (Mood=0)"          intercept 1 sexmf 0
/TEST = "Intercept: Women (Mood=0)"        intercept 1 sexmf 1
/TEST = "Intercept: Women Diff (Mood=0)"   sexmf 1
/TEST = "Within-Person Mood Effect: Men"    TVnm0 1 TVnm0*sexmf 0
/TEST = "Within-Person Mood Effect: Women"  TVnm0 1 TVnm0*sexmf 1
/TEST = "Within-Person Mood Effect: Women Diff" TVnm0*sexmf 1
/TEST = "Between-Person Mood Effect: Men"    TVnm0 1 TVnm0*sexmf 0 PMnm0 1 PMnm0*sexmf 0
/TEST = "Between-Person Mood Effect: Women"  TVnm0 1 TVnm0*sexmf 1 PMnm0 1 PMnm0*sexmf 1
/TEST = "Between-Person Mood Effect: Women Diff" TVnm0*sexmf 1 PMnm0*sexmf 1
/TEST = "Contextual Mood Effect: Men"        PMnm0 1 PMnm0*sexMF 0
/TEST = "Contextual Mood Effect: Women"      PMnm0 1 PMnm0*sexMF 1
/TEST = "Contextual Mood Effect: Women Diff" PMnm0*sexMF 1.

```

```

* STATA Model 3c: SPSS Model 2c: Fixed Effects of Sex (0=M, 1=F) by GMC Negative Mood
xtmixed lglucAM c.TVnm0 c.PMnm0 c.sexmf c.TVnm0sexmf c.PMnm0sexmf, ///
    || id: , variance ml covariance(un) residuals(independent,t(day)),
    estat ic, n(207),
lincom 1*_cons + 0*sexmf          // intercept: men (mood=0)
lincom 1*_cons + 1*sexmf          // intercept: women (mood=0)
lincom 1*sexmf                    // intercept: women diff (mood=0)
lincom 1*TVnm0 + 0*TVnm0sexmf     // within-person mood effect: men
lincom 1*TVnm0 + 1*TVnm0sexmf     // within-person mood effect: women
lincom 1*TVnm0sexmf               // within-person mood effect: women diff
lincom 1*TVnm0 + 0*TVnm0sexmf + 1*PMnm0 + 0*PMnm0sexmf // between-person mood effect: men
lincom 1*TVnm0 + 1*TVnm0sexmf + 1*PMnm0 + 1*PMnm0sexmf // between-person mood effect: women
lincom 1*TVnm0sexmf + 1*PMnm0sexmf // between-person mood effect: women diff
lincom 1*PMnm0 + 0*PMnm0sexmf     // contextual mood effect: men
lincom 1*PMnm0 + 1*PMnm0sexmf     // contextual mood effect: women
lincom 1*PMnm0sexmf               // contextual mood effect: women diff

```

Covariance Parameter Estimates					
Cov Parm	Subject	Estimate	Standard Error	Z	Pr > Z
UN(1,1)	ID	0.06074	0.006118	9.93	<.0001
day	ID	0.03007	0.000678	44.35	<.0001

Information Criteria						
Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
-1988.1	8	-1972.1	-1972.0	-1961.3	-1945.4	-1937.4

Solution for Fixed Effects					
Effect	Estimate	Standard Error	DF	t Value	Pr > t
Intercept	4.9539	0.02734	207	181.21	<.0001
TVnm0	0.03119	0.005937	3942	5.25	<.0001
PMnm0	0.1684	0.04886	214	3.45	0.0007
sexmf	-0.03619	0.03626	207	-1.00	0.3194
TVnm0*sexmf	-0.03443	0.007743	3942	-4.45	<.0001
PMnm0*sexmf	-0.1505	0.06184	214	-2.43	0.0158

Estimates					
Label	Estimate	Standard Error	DF	t Value	Pr > t
Intercept: Men (Mood=0)	4.9539	0.02734	207	181.21	<.0001
Intercept: Women (Mood=0)	4.9177	0.02382	207	206.42	<.0001
Intercept: Women Diff (Mood=0)	-0.03619	0.03626	207	-1.00	0.3194
Within-Person Effect: Men	0.03119	0.005937	3942	5.25	<.0001

Within-Person Effect: Women	-0.00325	0.004970	3942	-0.65	0.5138
Within-Person Effect: Women Diff	-0.03443	0.007743	3942	-4.45	<.0001
Between-Person Effect: Men	0.1996	0.04849	207	4.12	<.0001
Between-Person Effect: Women	0.01469	0.03759	207	0.39	0.6962
Between-Person Effect: Women Diff	-0.1849	0.06135	207	-3.01	0.0029
Contextual Effect: Men	0.1684	0.04886	214	3.45	0.0007
Contextual Effect: Women	0.01794	0.03790	214	0.47	0.6364
Contextual Effect: Women Diff	-0.1505	0.06184	214	-2.43	0.0158

Equations for each Conditional Model

2a) Predicting Glucose from Fixed Effects of Negative Mood using Person-Mean-Centering:

$$\text{Level 1: Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - \overline{\text{Mood}_i}) + e_i$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + U_{0i}$$

$$\text{Within-Person Mood: } \beta_{1i} = \gamma_{10}$$

2b) Testing random effect of WP negative mood under PMC:

$$\text{Level 1: Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - \overline{\text{Mood}_i}) + e_i$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + U_{0i}$$

$$\text{Within-Person Mood: } \beta_{1i} = \gamma_{10} + U_{1i}$$

2c) Adding moderation effects by sex (0=M, 1=F) for each mood effect under PMC:

$$\text{Level 1: Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - \overline{\text{Mood}_i}) + e_i$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + \gamma_{02} (\text{Woman}_i) + \gamma_{03} (\overline{\text{Mood}_i} - 0)(\text{Woman}_i) + U_{0i}$$

$$\text{Within-Person Mood: } \beta_{1i} = \gamma_{10} + \gamma_{12} (\text{Woman}_i)$$

3) Predicting Glucose from Time-Varying Negative Mood only (GMC):

$$\text{Level 1: Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - 0) + e_i$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Time-Varying Mood: } \beta_{1i} = \gamma_{10}$$

3a) Adding person mean negative mood at level 2 under GMC:

$$\text{Level 1: Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - 0) + e_i$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + U_{0i}$$

$$\text{Time-Varying Mood: } \beta_{1i} = \gamma_{10}$$

3b) Testing random effect of TV negative mood under GMC:

$$\text{Level 1: Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - 0) + e_i$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + U_{0i}$$

$$\text{Time-Varying Mood: } \beta_{1i} = \gamma_{10} + U_{1i}$$

3c) Adding main effect of sex and interactions with negative mood under GMC:

$$\text{Level 1: Glucose}_{ti} = \beta_{0i} + \beta_{1i} (\text{Mood}_{ti} - 0) + e_i$$

Level 2:

$$\text{Intercept: } \beta_{0i} = \gamma_{00} + \gamma_{01} (\overline{\text{Mood}_i} - 0) + \gamma_{02} (\text{Woman}_i) + \gamma_{03} (\overline{\text{Mood}_i} - 0)(\text{Woman}_i) + U_{0i}$$

$$\text{Time-Varying Mood: } \beta_{1i} = \gamma_{10} + \gamma_{12} (\text{Woman}_i)$$

Sample Results Section (note the order of the models is different than what is in this handout):

The effects of negative mood and sex on next day's morning glucose level were examined in 207 persons with type-2 diabetes over a 20-day period. Glucose was natural log transformed (after adding 1 to each score) to improve normality. Intraclass correlations as calculated from an empty (i.e., random intercept only) model were .69 for glucose and .39 for negative mood, such that 31% and 61% of the variance in each variable was within persons, over time, respectively. Preliminary analyses suggested that a random intercept only model for the variances of glucose over time had acceptable fit, and thus all conditional (predictor) models were examined using that structure as a baseline.

The time-varying (level-1) predictor for negative mood (left uncentered, given that 0 represented average level of the measure) was first entered into the model. A significant positive effect was obtained, such that higher daily levels of negative mood were related to higher daily levels of glucose. However, the inclusion of a single parameter for the effect of negative mood presumes that its between-person and within-person effects would be equivalent. This convergence hypothesis was tested explicitly by including person mean negative mood (also left uncentered, given that 0 represented average level of the original measure) as a level-2 predictor. The effect of person mean negative mood was significant, indicating that after controlling for absolute level of daily negative mood, persons with higher mean negative mood had higher mean glucose. Given that the significance of the level-2 effect also indicates that the between-person and within-person effects of negative mood were not equivalent, the model was re-specified to facilitate interpretation of these separate effects using group-mean-centering (i.e., person-mean-centering in this context). Specifically, a new level-1 predictor variable was created by subtracting each person's mean from daily negative mood, while the level-2 effect continued to be represented by the person mean. In this specification using person-mean-centering, the level-2 mean of negative mood represents the between-person effect directly and the level-1 within-person deviation of negative mood represents the within-person effect directly. Both the between- and within-person effects of negative mood were significantly positive. A random level-1 effect of negative mood was tested within both models, and was not found to be significant in either model, $-2\Delta LL (\sim 2) < 5.14, p > .05$, indicating no significant individual differences in the within-person effect of negative mood.

Three effects of sex were then entered into the person-mean-centered model, including a main effect of sex and interactions with the between-person and within-person effects of negative mood. The main effect of sex was non-significant, indicating no sex differences in mean glucose among persons with average levels of mean negative mood on average days (i.e., when average persons were at their mean). Given that both interactions were significant, however, results for both men and women will be presented as derived from ESTIMATE statements for the effects estimated specifically for each group within the overall model.

Parameters and fit statistics for this final model are given in Table 1. As shown, the intercept of 4.95 represents the expected morning LN glucose for a man with an average level of mean negative mood on an average day (i.e., both mean and person-mean-centered negative mood at 0). Men showed significant between-person and within-person effects of negative mood, such that for every unit higher in person mean negative mood, mean glucose was expected to be 0.20 higher (i.e., the between-person effect), and for every unit higher in negative mood on a given day relative to his own mean, glucose that next morning was expected to be 0.03 higher as well (i.e., the within-person effect). Thus, in men, being higher overall in negative mood and higher than usual in negative mood were each related to higher levels of glucose, and these effects were significantly different in magnitude (contextual mood effect = 0.17, SE = .05, $p < .001$). As shown in Figure 1, however, these patterns were not found in women, as indicated by the significant interactions with sex. Specifically, the between-person and within-person effects of negative mood in women were 0.015 (SE = 0.038) and -0.003 (SE = 0.005), respectively. Neither effect was significant nor did they differ significantly in magnitude (contextual mood effect = 0.018, SE = .038). Both effects of negative mood were significantly smaller than in men (interaction terms of sex with between-person and within-person negative mood of -0.185 and -0.034, respectively). Finally, the contextual effect of negative mood, or the difference between the between-person and within-person effects of negative mood, was significantly larger for men (0.151, SE = 0.062, $p = .016$).

(Table 1 would have all parameter estimates and fit statistics from final model, see chapter 8 for examples)
(Figure 1 would show the within-person effect of negative mood for men and women with low or high mean negative mood – see spreadsheet)